

xtdpd — Linear dynamic panel-data estimation

Syntax	Menu	Description	Options
Remarks and examples	Stored results	Methods and formulas	Acknowledgment
References	Also see		

Syntax

xtdpd *depvar* [*indepvars*] [*if*] [*in*], dgmmiv(*varlist* [...]) [*options*]

<i>options</i>	Description
Model	
* <u>d</u> gmmiv(<i>varlist</i> [...])	GMM-type instruments for the difference equation; can be specified more than once
<u>l</u> gmmiv(<i>varlist</i> [...])	GMM-type instruments for the level equation; can be specified more than once
iv(<i>varlist</i> [...])	standard instruments for the difference and level equations; can be specified more than once
div(<i>varlist</i> [...])	standard instruments for the difference equation only; can be specified more than once
liv(<i>varlist</i>)	standard instruments for the level equation only; can be specified more than once
<u>no</u> constant	suppress constant term
<u>two</u> step	compute the two-step estimator instead of the one-step estimator
<u>has</u> cons	check for collinearity only among levels of independent variables; by default checks occur among levels and differences
<u>f</u> odeviation	use forward-orthogonal deviations instead of first differences
SE/Robust	
vce(<i>vcetype</i>)	<i>vcetype</i> may be <u>g</u> mm or <u>r</u> obust
Reporting	
<u>level</u> (#)	set confidence level; default is <u>level</u> (95)
<u>artests</u> (#)	use # as maximum order for AR tests; default is <u>artests</u> (2)
<i>display_options</i>	control spacing and line width
<u>coef</u> legend	display legend instead of statistics

*dgmmiv() is required.

A panel variable and a time variable must be specified; use xtset; see [XT] xtset.

depvar, *indepvars*, and all *varlists* may contain time-series operators; see [U] 11.4.4 Time-series varlists.

by, statsby, and xi are allowed; see [U] 11.1.10 Prefix commands.

coeflegend does not appear in the dialog box.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Menu

Statistics > Longitudinal/panel data > Dynamic panel data (DPD) > Linear DPD estimation

Description

Linear dynamic panel-data models include p lags of the dependent variable as covariates and contain unobserved panel-level effects, fixed or random. By construction, the unobserved panel-level effects are correlated with the lagged dependent variables, making standard estimators inconsistent. **xtdpd** fits a dynamic panel-data model by using the Arellano–Bond (1991) or the Arellano–Bover/Blundell–Bond (1995, 1998) estimator.

At the cost of a more complicated syntax, **xtdpd** can fit models with low-order moving-average correlation in the idiosyncratic errors or predetermined variables with a more complicated structure than allowed for **xtabond** or **xtdpdsys**; see [XT] **xtabond** and [XT] **xtdpdsys**.

Options

Model

dgm**miv**(*varlist* [, **lagrange**(*flag* [*llag*])) specifies GMM-type instruments for the differenced equation. Levels of the variables are used to form GMM-type instruments for the difference equation. All possible lags are used, unless **lagrange**(*flag llag*) restricts the lags to begin with *flag* and end with *llag*. You may specify as many sets of GMM-type instruments for the differenced equation as you need within the standard Stata limits on matrix size. Each set may have its own *flag* and *llag*. **dgm****miv**() is required.

lgm**miv**(*varlist* [, **lag**(#)]) specifies GMM-type instruments for the level equation. Differences of the variables are used to form GMM-type instruments for the level equation. The first lag of the differences is used unless **lag**(#) is specified, indicating that #th lag of the differences be used. You may specify as many sets of GMM-type instruments for the level equation as you need within the standard Stata limits on matrix size. Each set may have its own *lag*.

iv(*varlist* [, **nodifference**]) specifies standard instruments for both the differenced and level equations. Differences of the variables are used as instruments for the differenced equations, unless **nodifference** is specified, which requests that levels be used. Levels of the variables are used as instruments for the level equations. You may specify as many sets of standard instruments for both the differenced and level equations as you need within the standard Stata limits on matrix size.

div(*varlist* [, **nodifference**]) specifies additional standard instruments for the differenced equation. Specified variables may not be included in **iv**() or in **liv**(). Differences of the variables are used, unless **nodifference** is specified, which requests that levels of the variables be used as instruments for the differenced equation. You may specify as many additional sets of standard instruments for the differenced equation as you need within the standard Stata limits on matrix size.

liv(*varlist*) specifies additional standard instruments for the level equation. Specified variables may not be included in **iv**() or in **div**(). Levels of the variables are used as instruments for the level equation. You may specify as many additional sets of standard instruments for the level equation as you need within the standard Stata limits on matrix size.

noconstant; see [R] **estimation options**.

twostep specifies that the two-step estimator be calculated.

hascons specifies that **xtdpd** check for collinearity only among levels of independent variables; by default checks occur among levels and differences.

fodeviation specifies that forward-orthogonal deviations are to be used instead of first differences. **fodeviation** is not allowed when there are gaps in the data or when **lgmiv()** is specified.

SE/Robust

vce(vcetype) specifies the type of standard error reported, which includes types that are derived from asymptotic theory and that are robust to some kinds of misspecification; see [Methods and formulas](#).

vce(gmm), the default, uses the conventionally derived variance estimator for generalized method of moments estimation.

vce(robust) uses the robust estimator. For the one-step estimator, this is the Arellano–Bond robust VCE estimator. For the two-step estimator, this is the [Windmeijer \(2005\)](#) WC-robust estimator.

Reporting

level(#); see [\[R\] estimation options](#).

artests(#) specifies the maximum order of the autocorrelation test to be calculated. The tests are reported by **estat abond**; see [\[XT\] xtdpd postestimation](#). Specifying the order of the highest test at estimation time is more efficient than specifying it to **estat abond**, because **estat abond** must refit the model to obtain the test statistics. The maximum order must be less than or equal to the number of periods in the longest panel. The default is **artests(2)**.

display_options: **vsquish** and **nolstretch**; see [\[R\] estimation options](#).

The following option is available with **xtdpd** but is not shown in the dialog box:

coeflegend; see [\[R\] estimation options](#).

Remarks and examples

[stata.com](http://www.stata.com)

If you have not read [\[XT\] xtabond](#) and [\[XT\] xtdpdsys](#), you should do so before continuing.

Consider the dynamic panel-data model

$$y_{it} = \sum_{j=1}^p \alpha_j y_{i,t-j} + \mathbf{x}_{it} \boldsymbol{\beta}_1 + \mathbf{w}_{it} \boldsymbol{\beta}_2 + \nu_i + \epsilon_{it} \quad i = \{1, \dots, N\}; \quad t = \{1, \dots, T_i\} \quad (1)$$

where

the $\alpha_1, \dots, \alpha_p$ are p parameters to be estimated,

$\mathbf{x}_{it}$ is a $1 \times k_1$ vector of strictly exogenous covariates,

$\boldsymbol{\beta}_1$ is a $k_1 \times 1$ vector of parameters to be estimated,

$\mathbf{w}_{it}$ is a $1 \times k_2$ vector of predetermined covariates,

$\boldsymbol{\beta}_2$ is a $k_2 \times 1$ vector of parameters to be estimated,

ν_i are the panel-level effects (which may be correlated with x_{it} or w_{it}), and

and ϵ_{it} are i.i.d. or come from a low-order moving-average process, with variance σ_ϵ^2 .

Building on the work of [Anderson and Hsiao \(1981, 1982\)](#) and [Holtz-Eakin, Newey, and Rosen \(1988\)](#), [Arellano and Bond \(1991\)](#) derived one-step and two-step GMM estimators using moment conditions in which lagged levels of the dependent and predetermined variables were instruments for the differenced equation. [Blundell and Bond \(1998\)](#) show that the lagged-level instruments in the Arellano–Bond estimator become weak as the autoregressive process becomes too persistent or the ratio of the variance of the panel-level effect ν_i to the variance of the idiosyncratic error ϵ_{it} becomes too large. Building on the work of [Arellano and Bover \(1995\)](#), [Blundell and Bond \(1998\)](#) proposed a system estimator that uses moment conditions in which lagged differences are used as instruments for the level equation in addition to the moment conditions of lagged levels as instruments for the differenced equation. The additional moment conditions are valid only if the initial condition $E[\nu_i \Delta y_{i2}] = 0$ holds for all i ; see [Blundell and Bond \(1998\)](#) and [Blundell, Bond, and Windmeijer \(2000\)](#).

`xtdpd` fits dynamic panel-data models by using the Arellano–Bond or the Arellano–Bover/Blundell–Bond system estimator. The parameters of many standard models can be more easily estimated using the Arellano–Bond estimator implemented in `xtabond` or using the Arellano–Bover/Blundell–Bond system estimator implemented in `xtdpdsys`; see [\[XT\] xtabond](#) and [\[XT\] xtdpdsys](#). `xtdpd` can fit more complex models at the cost of a more complicated syntax. That the idiosyncratic errors follow a low-order MA process and that the predetermined variables have a more complicated structure than accommodated by `xtabond` and `xtdpdsys` are two common reasons for using `xtdpd` instead of `xtabond` or `xtdpdsys`.

The standard GMM robust two-step estimator of the VCE is known to be seriously biased. [Windmeijer \(2005\)](#) derived a bias-corrected robust estimator for two-step VCEs from GMM estimators known as the WC-robust estimator, which is implemented in `xtdpd`.

The Arellano–Bond test of autocorrelation of order m and the Sargan test of overidentifying restrictions derived by [Arellano and Bond \(1991\)](#) are computed by `xtdpd` but reported by `estat abond` and `estat sargan`, respectively; see [\[XT\] xtdpd postestimation](#).

Because `xtdpd` extends `xtabond` and `xtdpdsys`, [\[XT\] xtabond](#) and [\[XT\] xtdpdsys](#) provide useful background.

► Example 1: An Arellano–Bond estimator

[Arellano and Bond \(1991\)](#) apply their new estimators and test statistics to a model of dynamic labor demand that had previously been considered by [Layard and Nickell \(1986\)](#), using data from an unbalanced panel of firms from the United Kingdom. All variables are indexed over the firm i and time t . In this dataset, n_{it} is the log of employment in firm i inside the United Kingdom at time t , w_{it} is the natural log of the real product wage, k_{it} is the natural log of the gross capital stock, and ys_{it} is the natural log of industry output. The model also includes time dummies `yr1980`, `yr1981`, `yr1982`, `yr1983`, and `yr1984`. To gain some insight into the syntax for `xtdpd`, we reproduce the first example from [\[XT\] xtabond](#) using `xtdpd`:

```
. use http://www.stata-press.com/data/r13/abdata
. xtdpd L(0/2).n L(0/1).w L(0/2).(k ys) yr1980-yr1984 year, noconstant
> div(L(0/1).w L(0/2).(k ys) yr1980-yr1984 year) dgmiv(n)

Dynamic panel-data estimation      Number of obs      =      611
Group variable: id                Number of groups   =      140
Time variable: year

                                Obs per group:   min =      4
                                                avg =  4.364286
                                                max =      6

Number of instruments =      41      Wald chi2(16)      =    1757.07
                                      Prob > chi2        =     0.0000
```

One-step results

	n	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
	n						
	L1.	.6862261	.1486163	4.62	0.000	.3949435	.9775088
	L2.	-.0853582	.0444365	-1.92	0.055	-.1724523	.0017358
	w						
	--.	-.6078208	.0657694	-9.24	0.000	-.7367265	-.4789151
	L1.	.3926237	.1092374	3.59	0.000	.1785222	.6067251
	k						
	--.	.3568456	.0370314	9.64	0.000	.2842653	.4294259
	L1.	-.0580012	.0583051	-0.99	0.320	-.172277	.0562747
	L2.	-.0199475	.0416274	-0.48	0.632	-.1015357	.0616408
	ys						
	--.	.6085073	.1345412	4.52	0.000	.3448115	.8722031
	L1.	-.7111651	.1844599	-3.86	0.000	-1.0727	-.3496304
	L2.	.1057969	.1428568	0.74	0.459	-.1741974	.3857912
	yr1980	.0029062	.0212705	0.14	0.891	-.0387832	.0445957
	yr1981	-.0404378	.0354707	-1.14	0.254	-.1099591	.0290836
	yr1982	-.0652767	.048209	-1.35	0.176	-.1597646	.0292111
	yr1983	-.0690928	.0627354	-1.10	0.271	-.1920521	.0538664
	yr1984	-.0650302	.0781322	-0.83	0.405	-.2181665	.0881061
	year	.0095545	.0142073	0.67	0.501	-.0182912	.0374002

Instruments for differenced equation

GMM-type: L(2/.)n

Standard: D.w LD.w D.k LD.k L2D.k D.ys LD.ys L2D.ys D.yr1980
D.yr1981 D.yr1982 D.yr1983 D.yr1984 D.year

Unlike most instrumental-variables estimation commands, the independent variables in the varlist are not automatically used as instruments. In this example, all the independent variables are strictly exogenous, so we include them in `div()`, a list of variables whose first differences will be instruments for the differenced equation. We include the dependent variable in `dgmiv()`, a list of variables whose lagged levels will be used to create GMM-type instruments for the differenced equation. (GMM-type instruments are discussed in a [technical note](#) below.)

The footer in the output reports the instruments used. The first line indicates that `xtdpd` used lags from 2 on back to create the GMM-type instruments described in [Arellano and Bond \(1991\)](#) and [Holtz-Eakin, Newey, and Rosen \(1988\)](#). The second line says that the first difference of all the variables included in the `div()` varlist were used as standard instruments for the differenced equation.

□ **Technical note**

GMM-type instruments are built from lags of one variable. Ignoring the strictly exogenous variables for simplicity, our model is

$$n_{it} = \alpha_1 n_{it-1} + \alpha_2 n_{it-2} + \nu_i + \epsilon_{it} \tag{2}$$

After differencing we have

$$\Delta n_{it} = \Delta \alpha_1 n_{it-1} + \Delta \alpha_2 n_{it-2} + \Delta \epsilon_{it} \tag{3}$$

Equation (3) implies that we need instruments that are not correlated with either ϵ_{it} or ϵ_{it-1} . Equation (2) shows that L2.n is the first lag of n that is not correlated with ϵ_{it} or ϵ_{it-1} , so it is the first lag of n that can be used to instrument the differenced equation.

Consider the following data from one of the complete panels in the [previous example](#):

```
. list id year n L2.n dL2.n if id==140
```

	id	year	n	L2. n	L2D. n
1023.	140	1976	.4324315	.	.
1024.	140	1977	.3694925	.	.
1025.	140	1978	.3541718	.4324315	.
1026.	140	1979	.3632532	.3694925	-.0629391
1027.	140	1980	.3371863	.3541718	-.0153207
1028.	140	1981	.285179	.3632532	.0090815
1029.	140	1982	.1756326	.3371863	-.026067
1030.	140	1983	.1275133	.285179	-.0520073
1031.	140	1984	.0889263	.1756326	-.1095464

The missing values in L2D.n show that we lose 3 observations because of lags and the difference that removes the panel-level effects. The first nonmissing observation occurs in 1979 and observations on n from 1976 and 1977 are available to instrument the 1979 differenced equation. The table below gives the observations available to instrument the differenced equation for the data above.

Year of difference errors	Years of instruments	Number of instruments
1979	1976–1977	2
1980	1976–1978	3
1981	1976–1979	4
1982	1976–1980	5
1983	1976–1981	6
1984	1976–1982	7

The table shows that there are a total of 27 GMM-type instruments.

The output in the example above informs us that there were a total of 41 instruments applied to the differenced equation. Because there are 14 standard instruments, there must have been 27 GMM-type instruments, which matches our above calculation.



► Example 2: An Arellano–Bond estimator with predetermined variables

Sometimes we cannot assume strict exogeneity. Recall that a variable x_{it} is said to be strictly exogenous if $E[x_{it}\epsilon_{is}] = 0$ for all t and s . If $E[x_{it}\epsilon_{is}] \neq 0$ for $s < t$ but $E[x_{it}\epsilon_{is}] = 0$ for all $s \geq t$, the variable is said to be predetermined. Intuitively, if the error term at time t has some feedback on the subsequent realizations of x_{it} , x_{it} is a predetermined variable. In the output below, we use `xtdpd` to reproduce [example 6](#) in [\[XT\] xtabond](#).

```
. xtdpd L(0/2).n L(0/1).(w ys) L(0/2).k yr1980-yr1984 year,
> div(L(0/1).(ys) yr1980-yr1984 year) dgmmlv(n) dgmmlv(L.w L2.k, lag(1.))
> twostep noconstant vce(robust)
```

Dynamic panel-data estimation	Number of obs	=	611
Group variable: id	Number of groups	=	140
Time variable: year			
	Obs per group:	min =	4
		avg =	4.364286
		max =	6
Number of instruments =	83	Wald chi2(15)	= 958.30
		Prob > chi2	= 0.0000

Two-step results

(Std. Err. adjusted for clustering on id)

n	Coef.	WC-Robust Std. Err.	z	P> z	[95% Conf. Interval]	
n						
L1.	.8580958	.1265515	6.78	0.000	.6100594	1.106132
L2.	-.081207	.0760703	-1.07	0.286	-.2303022	.0678881
w						
--.	-.6910855	.1387684	-4.98	0.000	-.9630666	-.4191044
L1.	.5961712	.1497338	3.98	0.000	.3026982	.8896441
ys						
--.	.6936392	.1728623	4.01	0.000	.3548354	1.032443
L1.	-.8773678	.2183085	-4.02	0.000	-1.305245	-.449491
k						
--.	.4140654	.1382788	2.99	0.003	.1430439	.6850868
L1.	-.1537048	.1220244	-1.26	0.208	-.3928681	.0854586
L2.	-.1025833	.0710886	-1.44	0.149	-.2419143	.0367477
yr1980	-.0072451	.017163	-0.42	0.673	-.0408839	.0263938
yr1981	-.0609608	.030207	-2.02	0.044	-.1201655	-.0017561
yr1982	-.1130369	.0454826	-2.49	0.013	-.2021812	-.0238926
yr1983	-.1335249	.0600213	-2.22	0.026	-.2511645	-.0158853
yr1984	-.1623177	.0725434	-2.24	0.025	-.3045001	-.0201352
year	.0264501	.0119329	2.22	0.027	.003062	.0498381

Instruments for differenced equation

GMM-type: L(2/.)n L(1/.)L.w L(1/.)L2.k

Standard: D.ys LD.ys D.yr1980 D.yr1981 D.yr1982 D.yr1983 D.yr1984
D.year

The footer informs us that we are now including GMM-type instruments from the first lag of `L.w` on back and from the first lag of `L2.k` on back.

▶ Example 3: A weaker definition of predetermined variables

As discussed in [XT] xtabond and [XT] xtdpdsys, xtabond and xtdpdsys both use a strict definition of predetermined variables with lags. In the strict definition, the most recent lag of the variable in `pre()` is considered predetermined. (Here specifying `pre(w, lag(1, .))` to `xtabond` means that `L.w` is a predetermined variable and `pre(k, lag(2, .))` means that `L2.k` is a predetermined variable.) In a weaker definition, the current observation is considered predetermined, but subsequent lags are included in the model. Here `w` and `k` would be predetermined instead of `L.w` and `L2.w`. The output below implements this weaker definition for the [previous example](#).

```

. xtdpd L(0/2).n L(0/1).(w ys) L(0/2).k yr1980-yr1984 year,
> div(L(0/1).(ys) yr1980-yr1984 year) dgmminv(n) dgmminv(w k, lag(1 .))
> twostep noconstant vce(robust)

Dynamic panel-data estimation                Number of obs      =           611
Group variable: id                         Number of groups    =           140
Time variable: year

Obs per group:   min =              4
                  avg =     4.364286
                  max =              6

Number of instruments =      101             Wald chi2(15)        =      879.53
                                                Prob > chi2         =      0.0000

Two-step results
                                         (Std. Err. adjusted for clustering on id)

```

n	Coef.	WC-Robust Std. Err.	z	P> z	[95% Conf. Interval]	
n						
L1.	.6343155	.1221058	5.19	0.000	.3949925	.8736384
L2.	-.0871247	.0704816	-1.24	0.216	-.2252661	.0510168
w						
--.	-.720063	.1133359	-6.35	0.000	-.9421973	-.4979287
L1.	.238069	.1223186	1.95	0.052	-.0016712	.4778091
ys						
--.	.5999718	.1653036	3.63	0.000	.2759827	.923961
L1.	-.5674808	.1656411	-3.43	0.001	-.8921314	-.2428303
k						
--.	.3931997	.0986673	3.99	0.000	.1998153	.5865842
L1.	-.0019641	.0772814	-0.03	0.980	-.1534329	.1495047
L2.	-.0231165	.0487317	-0.47	0.635	-.1186288	.0723958
yr1980	-.006209	.0162138	-0.38	0.702	-.0379875	.0255694
yr1981	-.0398491	.0313794	-1.27	0.204	-.1013516	.0216535
yr1982	-.0525715	.0397346	-1.32	0.186	-.1304498	.0253068
yr1983	-.0451175	.051418	-0.88	0.380	-.145895	.05566
yr1984	-.0437772	.0614391	-0.71	0.476	-.1641955	.0766412
year	.0173374	.0108665	1.60	0.111	-.0039605	.0386352

```

Instruments for differenced equation
GMM-type: L(2/.).n L(1/.).w L(1/.).k
Standard: D.ys LD.ys D.yr1980 D.yr1981 D.yr1982 D.yr1983 D.yr1984
D.year

```

As expected, the output shows that the additional 18 instruments available under the weaker definition can affect the magnitudes of the estimates. Applying the stricter definition when the true model was generated by the weaker definition yielded consistent but inefficient results; there were some additional

moment conditions that could have been included but were not. In contrast, applying the weaker definition when the true model was generated by the stricter definition yields inconsistent estimates.

◀

► Example 4: A system estimator of a dynamic panel-data model

Here we use `xtdpd` to reproduce [example 2](#) from [\[XT\] xtdpdsys](#) in which we used the system estimator to fit a model with predetermined variables.

```
. xtdpd L(0/1).n L(0/2).(w k) yr1980-yr1984 year,
> div(yr1980-yr1984 year) dgmdiv(n) dgmdiv(L2.(w k), lag(1 .))
> lgmmiv(n L1.(w k)) vce(robust) hascons
```

Dynamic panel-data estimation	Number of obs	=	751
Group variable: id	Number of groups	=	140
Time variable: year			
	Obs per group:	min =	5
		avg =	5.364286
		max =	7
Number of instruments =	95	Wald chi2(13)	= 7562.80
		Prob > chi2	= 0.0000

One-step results

(Std. Err. adjusted for clustering on id)

n	Coef.	Robust Std. Err.	z	P> z	[95% Conf. Interval]	
n						
L1.	.913278	.0460602	19.83	0.000	.8230017	1.003554
w						
--.	-.728159	.1019044	-7.15	0.000	-.927888	-.5284301
L1.	.5602737	.1939617	2.89	0.004	.1801156	.9404317
L2.	-.0523028	.1487653	-0.35	0.725	-.3438775	.2392718
k						
--.	.4820097	.0760787	6.34	0.000	.3328983	.6311212
L1.	-.2846944	.0831902	-3.42	0.001	-.4477442	-.1216446
L2.	-.1394181	.0405709	-3.44	0.001	-.2189356	-.0599006
yr1980	-.0325146	.0216371	-1.50	0.133	-.0749226	.0098935
yr1981	-.0726116	.0346482	-2.10	0.036	-.1405207	-.0047024
yr1982	-.0477038	.0451914	-1.06	0.291	-.1362772	.0408696
yr1983	-.0396264	.0558734	-0.71	0.478	-.1491362	.0698835
yr1984	-.0810383	.0736648	-1.10	0.271	-.2254186	.063342
year	.0192741	.0145326	1.33	0.185	-.0092092	.0477574
_cons	-37.34972	28.77747	-1.30	0.194	-93.75253	19.05308

Instruments for differenced equation

GMM-type: L(2/.)n L(1/.)L2.w L(1/.)L2.k

Standard: D.yr1980 D.yr1981 D.yr1982 D.yr1983 D.yr1984 D.yr

Instruments for level equation

GMM-type: LD.n L2D.w L2D.k

Standard: _cons

The first lags of the variables included in `lgmmiv()` are used to create GMM-type instruments for the level equation. Only the first lags of the variables in `lgmmiv()` are used because the moment conditions using higher lags are redundant; see [Blundell and Bond \(1998\)](#) and [Blundell, Bond, and Windmeijer \(2000\)](#).

◀

➤ Example 5: Allowing for MA(1) errors

All the previous examples have used moment conditions that are valid only if the idiosyncratic errors are i.i.d. This example shows how to use `xtdpd` to estimate the parameters of a model with first-order moving-average [MA(1)] errors using the Arellano–Bond estimator, the Arellano–Bover/Blundell–Bond system estimator, or any other consistent GMM estimator you want to specify. For simplicity, we assume that the independent variables are strictly exogenous. Also, to highlight the fact that we can specify the instrument list flexibly, we only include the levels and first lags of the exogenous variables in the instrument list. An Arellano–Bond estimator, for instance, would have included levels and first and second lags of the exogenous variables.

We begin by noting that the Sargan test rejects the null hypothesis that the overidentifying restrictions are valid in the model with i.i.d. errors.

```
. xtdpd L(0/1).n L(0/2).(w k) yr1980-yr1984 year,
> div(L(0/1).(w k) yr1980-yr1984 year) dgmiv(n) hascons
(output omitted)

. estat sargan
Sargan test of overidentifying restrictions
H0: overidentifying restrictions are valid
chi2(24)      = 49.70094
Prob > chi2   = 0.0015
```

Assuming that the idiosyncratic errors are MA(1) implies that only lags three or higher are valid instruments for the differenced equation. (See the [technical note](#) below.)

```
. xtdpd L(0/1).n L(0/2).(w k) yr1980-yr1984 year,
> div(L(0/1).(w k) yr1980-yr1984 year) dgmiv(n, lag(3 .)) hascons

Dynamic panel-data estimation      Number of obs      =      751
Group variable: id                 Number of groups    =      140
Time variable: year

Obs per group:   min =      5
                  avg = 5.364286
                  max =      7

Number of instruments =      32      Wald chi2(13)      =    1195.04
                                      Prob > chi2        =      0.0000
```

One-step results

	n	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
n							
	L1.	.8696303	.2014473	4.32	0.000	.4748008	1.26446
w							
	--.	-.5802971	.0762659	-7.61	0.000	-.7297756	-.4308187
	L1.	.2918658	.1543883	1.89	0.059	-.0107296	.5944613
	L2.	-.5903459	.2995123	-1.97	0.049	-1.177379	-.0033126
k							
	--.	.3428139	.0447916	7.65	0.000	.2550239	.4306039
	L1.	-.1383918	.0825823	-1.68	0.094	-.3002502	.0234665
	L2.	-.0260956	.1535855	-0.17	0.865	-.3271177	.2749265
	yr1980	-.0036873	.0301587	-0.12	0.903	-.0627973	.0554226
	yr1981	.00218	.0592014	0.04	0.971	-.1138526	.1182125
	yr1982	.0782939	.0897622	0.87	0.383	-.0976367	.2542246
	yr1983	.1734231	.1308914	1.32	0.185	-.0831193	.4299655
	yr1984	.2400685	.1734456	1.38	0.166	-.0998787	.5800157
	year	-.0354681	.0309963	-1.14	0.253	-.0962198	.0252836
	_cons	73.13706	62.61443	1.17	0.243	-49.58496	195.8591

```
Instruments for differenced equation
      GMM-type: L(3/.)n
      Standard: D.w LD.w D.k LD.k D.yr1980 D.yr1981 D.yr1982 D.yr1983
                D.yr1984 D.year
Instruments for level equation
      Standard: _cons
```

The results from `estat sargan` no longer reject the null hypothesis that the overidentifying restrictions are valid.

```
. estat sargan
Sargan test of overidentifying restrictions
      H0: overidentifying restrictions are valid
      chi2(18)      = 20.80081
      Prob > chi2   = 0.2896
```

Moving on to the system estimator, we note that the Sargan test rejects the null hypothesis after fitting the model with i.i.d. errors.

```
. xtdpd L(0/1).n L(0/2).(w k) yr1980-yr1984 year,
> div(L(0/1).(w k) yr1980-yr1984 year) dgmiv(n) lgmmiv(n) hascons
(output omitted)
. estat sargan
Sargan test of overidentifying restrictions
      H0: overidentifying restrictions are valid
      chi2(31)      = 59.22907
      Prob > chi2   = 0.0017
```

Now we fit the model using the additional moment conditions constructed from the second lag of `n` as an instrument for the level equation.

```
. xtdpd L(0/1).n L(0/2).(w k) yr1980-yr1984 year,
> div(L(0/1).(w k) yr1980-yr1984 year) dgmiv(n, lag(3 .)) lgmmiv(n, lag(2))
> hascons

Dynamic panel-data estimation          Number of obs      =      751
Group variable: id                    Number of groups   =      140
Time variable: year

Obs per group:   min =      5
                  avg =  5.364286
                  max =      7

Number of instruments =      38          Wald chi2(13)      =    3680.01
                                           Prob > chi2       =      0.0000

One-step results
```

n	Coef.	Std. Err.	z	P> z	[95% Conf. Interval]	
n						
L1.	.9603675	.095608	10.04	0.000	.7729794	1.147756
w						
--.	-.5433987	.068835	-7.89	0.000	-.6783128	-.4084845
L1.	.4356183	.0881727	4.94	0.000	.262803	.6084336
L2.	-.2785721	.1115061	-2.50	0.012	-.4971201	-.0600241
k						
--.	.3139331	.0419054	7.49	0.000	.2317999	.3960662
L1.	-.160103	.0546915	-2.93	0.003	-.2672963	-.0529096
L2.	-.1295766	.0507752	-2.55	0.011	-.2290943	-.030059
yr1980	-.0200704	.0248954	-0.81	0.420	-.0688644	.0287236
yr1981	-.0425838	.0422155	-1.01	0.313	-.1253246	.040157
yr1982	.0048723	.0600938	0.08	0.935	-.1129093	.122654
yr1983	.0458978	.0785687	0.58	0.559	-.1080941	.1998897
yr1984	.0633219	.1026188	0.62	0.537	-.1378074	.2644511
year	-.0075599	.019059	-0.40	0.692	-.0449148	.029795
_cons	16.20856	38.00619	0.43	0.670	-58.28221	90.69932

```
Instruments for differenced equation
GMM-type: L(3/.)n
Standard: D.w LD.w D.k LD.k D.yr1980 D.yr1981 D.yr1982 D.yr1983
          D.yr1984 D.year
Instruments for level equation
GMM-type: L2D.n
Standard: _cons
```

The estimate of the coefficient on L.n is now .96. [Blundell, Bond, and Windmeijer \(2000, 63–65\)](#) show that the moment conditions in the system estimator remain informative as the true coefficient on L.n approaches unity. [Holtz-Eakin, Newey, and Rosen \(1988\)](#) show that because the large-sample distribution of the estimator is derived for fixed number of periods and a growing number of individuals there is no “unit-root” problem.

The results from `estat sargan` no longer reject the null hypothesis that the overidentifying restrictions are valid.

```
. estat sargan
Sargan test of overidentifying restrictions
H0: overidentifying restrictions are valid
chi2(24)      =  27.22585
Prob > chi2   =   0.2940
```

□ Technical note

To find the valid moment conditions for the model with MA(1) errors, we begin by writing the model

$$n_{it} = \alpha n_{it-1} + \beta x_{it} + \nu_i + \epsilon_{it} + \gamma \epsilon_{it-1}$$

where the ϵ_{it} are assumed to be i.i.d.

Because the composite error, $\epsilon_{it} + \gamma \epsilon_{it-1}$, is MA(1), only lags two or higher are valid instruments for the level equation, assuming the initial condition that $E[\nu_i \Delta n_{i2}] = 0$. The key to this point is that lagging the above equation two periods shows that ϵ_{it-2} and ϵ_{it-3} appear in the equation for n_{it-2} . Because the ϵ_{it} are i.i.d., n_{it-2} is a valid instrument for the level equation with errors $\nu_i + \epsilon_{it} + \gamma \epsilon_{it-1}$. (n_{it-2} will be correlated with n_{it-1} but uncorrelated with the errors $\nu_i + \epsilon_{it} + \gamma \epsilon_{it-1}$.) An analogous argument works for higher lags.

First-differencing the above equation yields

$$\Delta n_{it} = \alpha \Delta n_{it-1} + \beta \Delta x_{it} + \Delta \epsilon_{it} + \gamma \Delta \epsilon_{it-1}$$

Because ϵ_{it-2} is the farthest lag of ϵ_{it} that appears in the differenced equation, lags three or higher are valid instruments for the differenced composite errors. (Lagging the level equation three periods shows that only ϵ_{it-3} and ϵ_{it-4} appear in the equation for n_{it-3} , which implies that n_{it-3} is a valid instrument for the current differenced equation. An analogous argument works for higher lags.)

□

Stored results

xtdpd stores the following in `e()`:

Scalars

<code>e(N)</code>	number of observations
<code>e(N_g)</code>	number of groups
<code>e(df_m)</code>	model degrees of freedom
<code>e(g_min)</code>	smallest group size
<code>e(g_avg)</code>	average group size
<code>e(g_max)</code>	largest group size
<code>e(t_min)</code>	minimum time in sample
<code>e(t_max)</code>	maximum time in sample
<code>e(chi2)</code>	χ^2
<code>e(arm#)</code>	test for autocorrelation of order #
<code>e(artests)</code>	number of AR tests computed
<code>e(sig2)</code>	estimate of σ_ϵ^2
<code>e(rss)</code>	sum of squared differenced residuals
<code>e(sargan)</code>	Sargan test statistic
<code>e(rank)</code>	rank of <code>e(V)</code>
<code>e(zrank)</code>	rank of instrument matrix

Macros

<code>e(cmd)</code>	<code>xtdpd</code>
<code>e(cmdline)</code>	command as typed
<code>e(depvar)</code>	name of dependent variable
<code>e(twostep)</code>	<code>twostep</code> , if specified
<code>e(ivar)</code>	variable denoting groups
<code>e(tvar)</code>	variable denoting time within groups
<code>e(vce)</code>	<code>vcetype</code> specified in <code>vce()</code>
<code>e(vcetype)</code>	title used to label Std. Err.
<code>e(system)</code>	<code>system</code> , if system estimator
<code>e(hascons)</code>	<code>hascons</code> , if specified
<code>e(transform)</code>	specified transform
<code>e(datasignature)</code>	checksum from <code>datasignature</code>
<code>e(properties)</code>	<code>b V</code>
<code>e(estat_cmd)</code>	program used to implement <code>estat</code>
<code>e(predict)</code>	program used to implement <code>predict</code>
<code>e(marginsok)</code>	predictions allowed by <code>margins</code>

Matrices

<code>e(b)</code>	coefficient vector
<code>e(V)</code>	variance–covariance matrix of the estimators

Functions

<code>e(sample)</code>	marks estimation sample
------------------------	-------------------------

Methods and formulas

Consider dynamic panel-data models of the form

$$y_{it} = \sum_{j=1}^p \alpha_j y_{i,t-j} + \mathbf{x}_{it} \boldsymbol{\beta}_1 + \mathbf{w}_{it} \boldsymbol{\beta}_2 + \nu_i + \epsilon_{it}$$

where the variables are as defined as in (1).

$\mathbf{x}$ and $\mathbf{w}$ may contain lagged independent variables and time dummies.

Let $\mathbf{X}_{it}^L = (y_{i,t-1}, y_{i,t-2}, \dots, y_{i,t-p}, \mathbf{x}_{it}, \mathbf{w}_{it})$ be the $1 \times K$ vector of covariates for i at time t , where $K = p + k_1 + k_2$, p is the number of included lags, k_1 is the number of strictly exogenous variables in $\mathbf{x}_{it}$, and k_2 is the number of predetermined variables in $\mathbf{w}_{it}$. (The superscript L stands for levels.)

Now rewrite this relationship as a set of T_i equations for each individual,

$$\mathbf{y}_i^L = \mathbf{X}_i^L \boldsymbol{\delta} + \nu_i \boldsymbol{\iota}_i + \boldsymbol{\epsilon}_i$$

where T_i is the number of observations available for individual i ; $\mathbf{y}_i$, $\boldsymbol{\iota}_i$, and $\boldsymbol{\epsilon}_i$ are $T_i \times 1$, whereas $\mathbf{X}_i$ is $T_i \times K$.

The estimators use both the levels and a transform of the variables in the above equation. Denote the transformed variables by an $*$, so that $\mathbf{y}_i^*$ is the transformed $\mathbf{y}_i^L$ and $\mathbf{X}_i^*$ is the transformed $\mathbf{X}_i^L$. The transform may be either the first difference or the forward-orthogonal deviations (FOD) transform. The (i, t) th observation of the FOD transform of a variable $\mathbf{x}$ is given by

$$x_{it}^* = c_t \left\{ x_{it} - \frac{1}{T-t} (x_{it+1} + x_{it+2} + \dots + x_{iT}) \right\}$$

where $c_t^2 = (T-t)/(T-t+1)$ and T is the number of observations on $\mathbf{x}$; see [Arellano and Bover \(1995\)](#) and [Arellano \(2003\)](#).

Here we present the formulas for the Arellano–Bover/Blundell–Bond system estimator. The formulas for the Arellano–Bond estimator are obtained by setting the additional level matrices in the system estimator to null matrices.

Stacking the transformed and untransformed vectors of the dependent variable for a given i yields

$$\mathbf{y}_i = \begin{pmatrix} \mathbf{y}_i^* \\ \mathbf{y}_i^L \end{pmatrix}$$

Similarly, stacking the transformed and untransformed matrices of the covariates for a given i yields

$$\mathbf{X}_i = \begin{pmatrix} \mathbf{X}_i^* \\ \mathbf{X}_i^L \end{pmatrix}$$

$\mathbf{Z}_i$ is a matrix of instruments,

$$\mathbf{Z}_i = \begin{pmatrix} \mathbf{Z}_{di} & \mathbf{0} & \mathbf{D}_i & \mathbf{0} & \mathbf{I}_i^d \\ \mathbf{0} & \mathbf{Z}_{Li} & \mathbf{0} & \mathbf{L}_i & \mathbf{I}_i^L \end{pmatrix}$$

where $\mathbf{Z}_{di}$ is the matrix of GMM-type instruments created from the `dgmiv()` options, $\mathbf{Z}_{Li}$ is the matrix of GMM-type instruments created from the `lgmiv()` options, $\mathbf{D}_i$ is the matrix of standard instruments created from the `div()` options, $\mathbf{L}_i$ is the matrix of standard instruments created from the `liv()` options, $\mathbf{I}_i^d$ is the matrix of standard instruments created from the `iv()` options for the differenced errors, and $\mathbf{I}_i^L$ is the matrix of standard instruments created from the `iv()` options for the level errors.

`div()`, `liv()`, and `iv()` simply add columns to instrument matrix. The GMM-type instruments are more involved. Begin by considering a simple balanced-panel example in which our model is

$$y_{it} = \alpha_1 y_{i,t-1} + \alpha_2 y_{i,t-2} + \nu_i + \epsilon_{it}$$

We do not need to consider covariates because strictly exogenous variables are handled using `div()`, `iv()`, or `liv()`, and predetermined or endogenous variables are handled analogous to the dependent variable.

Assume that the data come from a balanced panel in which there are no missing values. After first-differencing the equation, we have

$$\Delta y_{it} = \alpha_1 \Delta y_{i,t-1} + \alpha_2 \Delta y_{i,t-2} + \Delta \epsilon_{it}$$

The first 3 observations are lost to lags and differencing. If we assume that the ϵ_{it} are not autocorrelated, for each i at $t = 4$, y_{i1} and y_{i2} are valid instruments for the differenced equation. Similarly, at $t = 5$, y_{i1} , y_{i2} , and y_{i3} are valid instruments. We specify `dgmiv(y)` to obtain an instrument matrix with one row for each period that we are instrumenting:

$$\mathbf{Z}_{di} = \begin{pmatrix} y_{i1} & y_{i2} & 0 & 0 & 0 & \dots & 0 & 0 & 0 \\ 0 & 0 & y_{i1} & y_{i2} & y_{i3} & \dots & 0 & 0 & 0 \\ \vdots & \vdots & \vdots & \vdots & \ddots & \vdots & \vdots & \vdots & \vdots \\ 0 & 0 & 0 & 0 & \dots & 0 & y_{i1} & \dots & y_{i,T-2} \end{pmatrix}$$

Because $p = 2$, $\mathbf{Z}_{di}$ has $T - p - 1$ rows and $\sum_{m=p}^{T-2} m$ columns.

Specifying `lgmmiv(y)` creates the instrument matrix

$$\mathbf{Z}_{Li} = \begin{pmatrix} \Delta.y_{i2} & 0 & 0 & \dots & 0 \\ 0 & \Delta.y_{i3} & 0 & \dots & 0 \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ 0 & 0 & 0 & \dots & \Delta.y_{i(T_i-1)} \end{pmatrix}$$

This extends to other lag structures with complete data. Unbalanced data and missing observations are handled by dropping the rows for which there are no data and filling in zeros in columns where missing data are required. Suppose that, for some i , the $t = 1$ observation was missing but was not missing for some other panels. `dgmmiv(y)` would then create the instrument matrix

$$\mathbf{Z}_{di} = \begin{pmatrix} 0 & 0 & 0 & y_{i2} & y_{i3} & 0 & 0 & 0 & 0 & \dots & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & y_{i2} & y_{i3} & 0 & \dots & 0 & 0 & 0 \\ \vdots & \vdots & \vdots & \vdots & \vdots & \vdots & \vdots & \vdots & \ddots & \vdots & \vdots & \vdots & \vdots \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \dots & 0 & y_{i2} & \dots & y_{iT-2} \end{pmatrix}$$

$\mathbf{Z}_{di}$ has $T_i - p - 1$ rows and $\sum_{m=p}^{\tau-2} m$ columns, where $\tau = \max_i \tau_i$ and τ_i is the number of nonmissing observations in panel i .

After defining

$$\mathbf{Q}_{xz} = \sum_i \mathbf{X}'_i \mathbf{Z}_i$$

$$\mathbf{Q}_{zy} = \sum_i \mathbf{Z}'_i \mathbf{y}_i$$

$$\mathbf{W}_1 = \mathbf{Q}_{xz} \mathbf{A}_1 \mathbf{Q}'_{xz}$$

$$\mathbf{A}_1 = \left(\sum_i \mathbf{Z}'_i \mathbf{H}_{1i} \mathbf{Z}_i \right)^{-1}$$

and

$$\mathbf{H}_{1i} = \begin{pmatrix} \mathbf{H}_{di} & \mathbf{0} \\ \mathbf{0} & \mathbf{H}_{Li} \end{pmatrix}$$

the one-step estimates are given by

$$\hat{\beta}_1 = \mathbf{W}_1^{-1} \mathbf{Q}_{xz} \mathbf{A}_1 \mathbf{Q}_{zy}$$

When using the first-difference transform $\mathbf{H}_{di}$, is given by

$$\mathbf{H}_{di} = \begin{pmatrix} 1 & -.5 & 0 & \dots & 0 & 0 \\ -.5 & 1 & -.5 & \dots & 0 & 0 \\ \vdots & \vdots & \vdots & \ddots & \vdots & \vdots \\ 0 & 0 & 0 & \dots & 1 & -.5 \\ 0 & 0 & 0 & \dots & -.5 & 1 \end{pmatrix}$$

and $\mathbf{H}_{Li}$ is given by 0.5 times the identity matrix. When using the FOD transform, both $\mathbf{H}_{di}$ and $\mathbf{H}_{Li}$ are equal to the identity matrix.

The transformed one-step residuals are given by

$$\hat{\epsilon}_{1i}^* = \mathbf{y}_i^* - \hat{\beta}_1 \mathbf{X}_i^*$$

which are used to compute

$$\hat{\sigma}_1^2 = (1/(N - K)) \sum_i^N \hat{\epsilon}_{1i}^{*'} \hat{\epsilon}_{1i}^*$$

The GMM one-step VCE is then given by

$$\hat{V}_{\text{GMM}}[\hat{\beta}_1] = \hat{\sigma}_1^2 \mathbf{W}_1^{-1}$$

The one-step level residuals are given by

$$\hat{\epsilon}_{1i}^L = \mathbf{y}_i^L - \hat{\beta}_1 \mathbf{X}_i^L$$

Stacking the residual vectors yields

$$\hat{\epsilon}_{1i} = \begin{pmatrix} \hat{\epsilon}_{1i}^* \\ \hat{\epsilon}_{1i}^L \end{pmatrix}$$

which is used to compute $\mathbf{H}_{2i} = \hat{\epsilon}_{1i}' \hat{\epsilon}_{1i}$, which is used in

$$\mathbf{A}_2 = \left(\sum_i \mathbf{Z}_i' \mathbf{H}_{2i} \mathbf{Z}_i \right)^{-1}$$

and the robust one-step VCE is given by

$$\hat{V}_{\text{robust}}[\hat{\beta}_1] = \mathbf{W}_1^{-1} \mathbf{Q}_{xz} \mathbf{A}_1 \mathbf{A}_2^{-1} \mathbf{A}_1 \mathbf{Q}_{xz}' \mathbf{W}_1^{-1}$$

$\hat{V}_{\text{robust}}[\hat{\beta}_1]$ is robust to heteroskedasticity in the errors.

After defining

$$\mathbf{W}_2 = \mathbf{Q}_{xz} \mathbf{A}_2 \mathbf{Q}'_{xz}$$

the two-step estimates are given by

$$\hat{\beta}_2 = \mathbf{W}_2^{-1} \mathbf{Q}_{xz} \mathbf{A}_2 \mathbf{Q}_{zy}$$

The GMM two-step VCE is then given by

$$\hat{V}_{\text{GMM}}[\hat{\beta}_2] = \mathbf{W}_2^{-1}$$

The GMM two-step VCE is known to be severely biased. [Windmeijer \(2005\)](#) derived the Windmeijer bias-corrected (WC) estimator for the robust VCE of two-step GMM estimators. `xtdpd` implements this WC-robust estimator of the VCE. The formulas for this method are involved; see [Windmeijer \(2005\)](#). The WC-robust estimator of the VCE is robust to heteroskedasticity in the errors.

Acknowledgment

We thank David Roodman of the Center for Global Development, who wrote `xtabond2`.

References

- Anderson, T. W., and C. Hsiao. 1981. Estimation of dynamic models with error components. *Journal of the American Statistical Association* 76: 598–606.
- . 1982. Formulation and estimation of dynamic models using panel data. *Journal of Econometrics* 18: 47–82.
- Arellano, M. 2003. *Panel Data Econometrics*. Oxford: Oxford University Press.
- Arellano, M., and S. Bond. 1991. Some tests of specification for panel data: Monte Carlo evidence and an application to employment equations. *Review of Economic Studies* 58: 277–297.
- Arellano, M., and O. Bover. 1995. Another look at the instrumental variable estimation of error-components models. *Journal of Econometrics* 68: 29–51.
- Baltagi, B. H. 2013. *Econometric Analysis of Panel Data*. 5th ed. Chichester, UK: Wiley.
- Blackburne, E. F., III, and M. W. Frank. 2007. Estimation of nonstationary heterogeneous panels. *Stata Journal* 7: 197–208.
- Blundell, R., and S. Bond. 1998. Initial conditions and moment restrictions in dynamic panel data models. *Journal of Econometrics* 87: 115–143.
- Blundell, R., S. Bond, and F. Windmeijer. 2000. Estimation in dynamic panel data models: Improving on the performance of the standard GMM estimator. In *Nonstationary Panels, Cointegrating Panels and Dynamic Panels*, ed. B. H. Baltagi, 53–92. New York: Elsevier.
- Bruno, G. S. F. 2005. Estimation and inference in dynamic unbalanced panel-data models with a small number of individuals. *Stata Journal* 5: 473–500.
- Hansen, L. P. 1982. Large sample properties of generalized method of moments estimators. *Econometrica* 50: 1029–1054.
- Holtz-Eakin, D., W. K. Newey, and H. S. Rosen. 1988. Estimating vector autoregressions with panel data. *Econometrica* 56: 1371–1395.
- Layard, R., and S. J. Nickell. 1986. Unemployment in Britain. *Economica* 53: S121–S169.
- Windmeijer, F. 2005. A finite sample correction for the variance of linear efficient two-step GMM estimators. *Journal of Econometrics* 126: 25–51.

Also see

- [XT] **xtdpd postestimation** — Postestimation tools for xtdpd
- [XT] **xtset** — Declare data to be panel data
- [XT] **xtabond** — Arellano–Bond linear dynamic panel-data estimation
- [XT] **xtdpdsys** — Arellano–Bover/Blundell–Bond linear dynamic panel-data estimation
- [XT] **xtivreg** — Instrumental variables and two-stage least squares for panel-data models
- [XT] **xtreg** — Fixed-, between-, and random-effects and population-averaged linear models
- [XT] **xtregar** — Fixed- and random-effects linear models with an AR(1) disturbance
- [R] **gmm** — Generalized method of moments estimation
- [U] **20 Estimation and postestimation commands**